Spreadsheets 2 - Functions and Charts

Lecture 12 – COMPSCI111/111G SS 2019

Today’s lecture

- IF function recap
- VLOOKUP and HLOOKUP
- Sorting data
- Inserting chart

IF functions

- **Makes a decision**
  - Different values used in the cell depending on the logical test

- **IF(logical_test , value_if_true, value_if_false)**



Example - coffee data

- **Imagine an experiment where we record the number of cups of coffee that we drink, and whether it was morning or afternoon. The table of data might appear as shown below:**

	A	B
1	**Cups of Coffee**	**AM/PM**
2	3	am
3	1	pm
4	2	am
5	1	am
6	3	pm
7	5	am
8	1	pm

- **How can we calculate the average number of coffees that we drink in the morning?**

Example - coffee data

- **Add a new column to store the morning coffee data.**
  - If the contents of column B is the text "am" then use the value stored in column A. Otherwise, leave it blank.

	A	B	C
1	**Cups Of Coffee**	**AM/PM**	**Morning**
2	3	am	3
3	1	pm	
4	2	am	2
5	1	am	1
6	3	pm	
7	5	am	5
8	1	pm	
9	***Average***		2.8

```
=IF(B2="am", A2, "")
```

IF function

- Inserts a value in a cell based on the outcome of a logical test (ie. true/false)
- Syntax:
  `=IF(logical_test, value_if_true, value_if_false)`
- Logical tests:
  - Comparison operators: =, <, >, >=, <=
  - Logical functions:
    - AND(a, b)
    - OR(a, b)
    - NOT(a)

IF function

- Use an IF function in cell D5 to check whether a child is under the maximum height *and* weight. If they are, write “Yes!”, otherwise write “No”. Ensure that your formula can be filled down.

	A	B	C	D
1	Max height:	1.2	metres	
2	Max weight:	40	kg	
3				
4	**Name**	**Height**	**Weight**	**Allowed in playground?**
5	Tom	1.45	56	No
6	Charlie	1.10	44	No
7	Ben	1.19	35	Yes!

IF function

- Syntax:
  `=IF(logical_test, value_if_true, value_if_false)`
- `=IF(`**`AND(B5<B1, C5<B2)`**`, value_if_true, value_if_false)`

	A	B	C	D
1	Max height:	1.2	metres	
2	Max weight:	40	kg	
3				
4	**Name**	**Height**	**Weight**	**Allowed in playground?**
5	Tom	1.45	56	No
6	Charlie	1.10	44	No
7	Ben	1.19	35	Yes!

IF function

- Syntax:

  `=IF(logical_test, value_if_true, value_if_false)`

- `=IF(AND(B5<B1, C5<B2), "Yes!", "No")`

	A	B	C	D
1	Max height:	1.2	metres	
2	Max weight:	40	kg	
3				
4	**Name**	**Height**	**Weight**	**Allowed in playground?**
5	Tom	1.45	56	No
6	Charlie	1.10	44	No
7	Ben	1.19	35	Yes!

IF function

- Syntax:
  `=IF(logical_test, value_if_true, value_if_false)`
- `=IF(AND(B5<`**`$B$1`**`, C5<`**`$B$2`**`), "Yes!", "No")`

	A	B	C	D
1	Max height:	1.2	metres	
2	Max weight:	40	kg	
3				
4	**Name**	**Height**	**Weight**	**Allowed in playground?**
5	Tom	1.45	56	No
6	Charlie	1.10	44	No
7	Ben	1.19	35	Yes!

Exercise - Simple IF

- **Given the wind speed as shown in the table below, write the formula that would appear in cell C2. Note that a Gale Warning is issued when the wind speed exceeds 63 km/hr.**

	A	B	C
1	**Date**	**Wind Speed**	**Warning Issued**
2	1/01/2007	3	
3	2/01/2007	57	
4	3/01/2007	89	Gale Warning
5	4/01/2007	60	
6	5/01/2007	5	
7	6/01/2007	84	Gale Warning
8	7/01/2007	87	Gale Warning
9	8/01/2007	8	

```
=IF(B2>63,"Gale Warning", "")
```

Exercise - IF

IF less than 50 percent of tickets available at a venue were sold, then the venue is too large. To produce the result in cell F7, what formula should you use in this cell?

	A	B	C	D	E	F
1		Ticket Sales				
2						
3	**Price**	$10.00				
4						
5	**Event**	**Tickets Available**	**Tickets Sold**	**Remaining**	**Sales**	**Venue too large?**
6	Cycling	4000	2000	2000	$20,000.00	No
7	Weightlifting	2000	750	1250	$7,500.00	Yes
8	Triathlon	1000	100	900	$1,000.00	Yes
9	Football	3000	3000	0	$30,000.00	No
10	Badminton	5000	4500	500	$45,000.00	No
11		15000	10350	4650	$103,500.00	

```
=IF(C7/B7<0.5, "Yes", "No")
```

Exercise – IF with a Boolean Function

- **Ticket Sales**
  - Check if more than 90% of the tickets were sold, or if less than 50% of the tickets were sold. In either case, a new venue is required next time.

	A	B	C	D	E	F	G
1		**Ticket Sales**					
2							
3	**Price**	$10.00					
4							
5	**Event**	**Tickets Available**	**Tickets Sold**	**Remaining**	**Sales**	**Venue too large?**	**New venue?**
6	Cycling	4000	2000	2000	$20,000.00	No	No
7	Weightlifting	2000	750	1250	$7,500.00	Yes	Yes
8	Triathlon	1000	100	900	$1,000.00	Yes	Yes
9	Football	3000	3000	0	$30,000.00	No	Yes
10	Badminton	5000	4500	500	$45,000.00	No	No
11		15000	10350	4650	$103,500.00		

```
=IF(OR(C9/B9>0.9,C9/B9<0.5),"Yes","No")
```

Lookup functions

- Sometimes we will need to look up values in a table in our spreadsheet
  - For example, matching a student's ID number with their name
- Two kinds of look up functions
  - VLOOKUP: used with vertical

	A	B
1	**Name**	**Age**
2	Eddie	18
3	Frank	21
4	Josh	24

  - HLOOKUP: used with horizontal

	A	B	C	D
6	Timing	1:15	1:20	1:25
7	Score	1	2	4

VLOOKUP

VLOOKUP(value, table, column, [range])

Value.

This is the value we already have written down. We want to use this value to look up a corresponding value in a table.

Range of cells.

This is the table we are using to look up the value in.

Usually we want to use absolute references for the table.

Number.

This specifies which column in the table contains the data we want.

Boolean value.

True if we want to match a range of values

False if we want an exact match.

VLOOKUP

- Syntax:
  `VLOOKUP(value, table, column,range)`
- **Value:** the cell that you are looking up
- **Table:** a range of cells containing the table, usually written as absolute references
- **Column:** the column of the table that contains the values we want to retrieve
- **Range:** this is a Boolean value; true if the lookup value falls within a range, false if an exact match is required

Example

- Use VLOOKUP to insert the students' surnames in the blank cells, given their ID

	A	B	C	D	E	F
1	**Students Enrolled**					
2	**ID**	**Surname**		**ID**	**First name**	**Surname**
3	5			5	Debra	Martinez
4	7			6	Ida	Holloway
5	19			7	Flora	Taylor
6				10	Jared	Shelton
7				11	Marianne	Fuller
8				12	Everett	Hill
9				16	Edward	Brown
10				19	Benny	Gibbs

VLOOKUP Example

	A	B	C	D	E	F
1	**Students Enrolled**					
2	**ID**	**Surname**		**ID**	**First name**	**Surname**
3	11	Fuller		5	Debra	Martinez
4	16	Brown		6	Ida	Holloway
5	5	Martinez		7	Flora	Taylor
6				10	Jared	Shelton
7				11	Marianne	Fuller
8				12	Everett	Hill
9				16	Edward	Brown
10				19	Benny	Gibbs

=VLOOKUP(A3, $D$3:$F$10, 3, false)

=VLOOKUP(value, table, column, range)

Exercise: ThinkGeek T-Shirts

http://www.thinkgeek.com/

Exercise

- **What formulae should be used in cells D15, E15, F15 and F26?**

	A	B	C	D	E	F
1	**T-Shirt Sizes**				**T-Shirt Prints**	
2	**Size**	**Price**			**Code**	**Description**
3	S	$ 10.99			1001	2 + 2 = 5
4	M	$ 11.99			1010	geek inside
5	L	$ 12.99			1011	<BODY>
6	XL	$ 13.99			1100	man woman
7	XXL	$ 14.99			1101	obey gravity
8	XXXL	$ 15.99			1110	I'm blogging this
9					1111	Arrrrrggh...
10						
11						
12						
13	**Invoice**					
14	**Code**	**Size**	**Number**	**Description**	**Price**	**Cost**
15	1010	M	1	geek inside	$ 11.99	$ 11.99
16	1010	L	1	geek inside	$ 12.99	$ 12.99
17	1011	S	3	<BODY>	$ 10.99	$ 32.97
18	1110	XL	1	I'm blogging this	$ 13.99	$ 13.99
19	1001	XL	1	2 + 2 = 5	$ 13.99	$ 13.99
20	1101	M	2	obey gravity	$ 11.99	$ 23.98
21	1111	M	1	Arrrrrggh...	$ 11.99	$ 11.99
22						
23						
24						
25						
26					Total	$ 121.90

Exercise

- D15

```
=VLOOKUP(A15,$E$3:$F$9,2,FALSE)
```

- E15

```
=VLOOKUP(B15,$A$3:$B$8,2,FALSE)
```

- F15

```
=E15*C15
```

- F26

```
=SUM(F15:F21)
```

HLOOKUP

- **Same as VLOOKUP, but for horizontal tables**

HLOOKUP(value, table, *row*, [range])



Value.

This is the value we already have written down. We want to use this value to look up a corresponding value in a table.

Range of cells.

This is the table we are using to look up the value in.

Usually we want to use absolute references for the table.

Number.

This specifies which row in the table contains the data we want.

Boolean value.

True if we want to match a range of values

False if we want an exact match.

HLOOKUP

- Same syntax as VLOOKUP, except it is used to look up values in horizontal tables
- Write a formula for C6 that finds the cost of tickets on a day in A6:A8 and multiplies the cost with the number of tickets

	A	B	C	D	E	F	G	H
1	**Day**	Mon	Tue	Wed	Thu	Fri	Sat	Sun
2	**Price**	$11	$11	$11	$15	$14	$13	$15
3								
4	**Tickets purchased**							
5	**Day**	**Tickets sold**	**Total cost**					
6	Mon	13	$143					
7	Thu	29	$435					
8	Sun	50	$750					

HLOOKUP

- Same syntax as VLOOKUP, except it is used to look up values in horizontal tables



	A	B	C	D	E	F	G	H
1	**Day**	Mon	Tue	Wed	Thu	Fri	Sat	Sun
2	**Price**	$11	$11	$11	$15	$14	$13	$15
3								
4	**Tickets purchased**							
5	**Day**	**Tickets sold**	**Total cost**					
6	Mon	13	$143					
7	Thu	29	$435					
8	Sun	50	$750					

=HLOOKUP(A6,$B$1:$H$2,2,FALSE)*B6

=HLOOKUP(value, table, row, range)

Exercise

- Write a formula in E3 that uses the table in cells A1 to B5 to find the person's grade and place it in the cell. Your formula must be able to be filled to the right

	A	B	C	D	E	F	G
1	**Score**	**Grade**		**Name**	**Tommy**	**Ben**	**Jemma**
2	0	Fail		**Score**	41	91	77
3	40	Pass		**Grade**	Pass	Perfect	Good
4	70	Good					
5	90	Perfect					

Exercise

- **=VLOOKUP(** , , , **)**

	A	B	C	D	E	F	G
1	**Score**	**Grade**		**Name**	**Tommy**	**Ben**	**Jemma**
2	0	Fail		**Score**	41	91	77
3	40	Pass		**Grade**	Pass	Perfect	Good
4	70	Good					
5	90	Perfect					

Exercise

- **What formula would be used in cell C7?**
  - Use a HLOOKUP

	A	B	C	D	E	F	G	H
1		Movie Prices						
2	**Day**	Mon	Tues	Wed	Thurs	Fri	Sat	Sun
3	**Price**	$11.00	$11.00	$11.00	$15.00	$15.00	$15.00	$15.00
4								
5	Movie Tickets							
6	**Name**	**Day**	**Cost**					
7	John	Tues	$11.00					
8	Jane	Thurs	$15.00					
9	Tom	Sat	$15.00					

```
=HLOOKUP(B7,$B$2:$H$3,2,FALSE)
```

Sorting data

- Excel can sort data using columns; Data → Sort



Sorting data

- When we click ‘OK’ the data is sorted

	A	B	C
1	**First name**	**Surname**	**Mark**
2	Michael	Diaz	100
3	Bruce	Carpenter	93
4	Rod	Smith	82
5	Mark	Evans	71
6	Gray	Redman	59
7	Matt	Fisher	47
8	Chris	Patterson	42
9	Tom	Smith	29
10	Colin	Stevens	21

Inserting a chart

- Once you have sorted data, you can create a Chart to insert in your spreadsheet
  - We'll use the data from the previous slide
- Decide on what is the best chart to use to present your data
- We also need to decide on the dependent and independent variable
  - Independent goes on the x-axis
  - Dependent goes on the y-axis

Inserting a chart

- Clicking on the 2D Column chart icon gives me a preview of my chart



Inserting a chart



Inserting a chart

- The chart that Excel generated had a few things missing



Inserting a chart

- Added axis title, adjusted scale, added trendline and equation



Summary

- Looked at three functions:
  - IF
  - VLOOKUP
  - HLOOKUP
- Discussed more Excel features:
  - Sorting data
  - Inserting and modifying charts